
ビジュアルヘルプ – グラフ（２） 

 

トレースの修正 

Graph メニューから Modify Trace Appearance を選択するか、グラフ内のトレースをダブルクリックすること

で、各トレースのグラフ上での表示方法を指定できます。 

画像プロットの場合、Modify Trace Appearance ではなく、Image メ

ニューから Modify Image Appearance を選択してください。 

 

コンタープロットでは、通常、Modify Contour Appearance を選択す

る必要があります。 

これを使って、すべてのコンターの表示を一度にコントロールします。 

ただし、１本の等高線を目立たせたい場合は、Modify Trace 

Appearance ダイアログを使います。 

 

修正するトレースの選択 

外観を変更したいトレースを Traces リストから選択してくださ

い。 

グラフ内のトレースをダブルクリックしてこのダイアログを表示

した場合、そのトレースは自動的に選択されます。 

複数のトレースを選択した場合、ダイアログ内の項目には最初に

選択したトレースの設定が表示されます。 

選択が完了したら、選択したトレースの設定を変更できます。 

この操作を行った後、別のトレースまたはトレースのセットを選

択し、さらに変更を加えることができます。 

Igorは行ったすべての変更を記憶するため、ダイアログを一度開

くだけで全ての操作を完了できます。 
 

 

表示モード 

選択したトレースの表示モードを Mode ポップアップメニューか

ら選択します。 

一般的に使われるモードは、Lines Between Points、Points、

Markers、Lines and Markers、Bars です。 

他のモードでは、Y=0 または別のトレースに対して、Sticks、

Bars、Fills を描画できます。 

マーカー 

Markers または Lines and Markers モードを選択した場合、マーカーの種類、マーカーのサイズ、マーカーの太

さ、およびマーカーを透明にするか色で塗りつぶすかを選択することもできます。 

マーカーのサイズは 0から 200までの小数です。 

0は「自動」を意味し、Igor がグラフのサイズに適したマーカーサイズを選択します。 

マーカーの太さは 0から 50ポイントまでの小数です。 

小数点は画面上では表示されない場合がありますが、エクスポートや印刷されたグラフィックでは確認できます。 

マーカーの太さを 0に設定すると、マーカーが消えます。 

マーカーのサイズと太さの間には相互作用があります。 

マーカーを対称にするために必要であれば、Igorはマーカーのサイズを調整します。 

マーカーの調整前の幅と高さは、マーカーサイズの設定値 sに対して 2*s+1ポイントです。 

マーカーと対応するマーカー番号の一覧表は以下の通りです。 

カスタムのマーカーを作成することもできます。 

SetWindow コマンドの markerHook キーワードを参照してください。 

 

マーカーの色 

Igor はグラフマーカーの３つの色（トレース色、ストローク色、塗りつぶし色）をコントロールします。 

トレース色は、トレース全体に対する選択された色であり、どのトレースモードでも同じです。 

ストローク色は、マーカーの輪郭線となる線の色です。 

デフォルトでは、ストローク色はトレース色と同じです。 

塗りつぶし色は、中空マーカーの背景領域を塗りつぶすために使われます。 

デフォルトでは塗りつぶし色は設定されておらず、中空マーカーの内部を通して背景オブジェクトが見えるようにな

っています。 

以下は、トレースカラーを黒、マーカーサイズを 10、マーカーのストローク太さを 2ポイントに設定した一部のマ

ーカーのサンプルです。 

 

青い長方形は、中空マーカー（5、11、12、8、41、42、43）の内部に透けて見えます。 

ソリッドマーカー（16、19）はトレース色が黒であるため、黒一色で表示されます。 

ストローク色は Modify Trace Appearance ダイアログで設定できます。 

Stroke チェックボックスをオンにして色を選択します。 

また、ModifyGraph mrkStrokeRGB コマンドも使用できます。 

次の図では、コマンドを使ってストロークの色を緑に設定しました： 

ModifyGraph mrkStrokeRGB=(1,52428,26586) 

 

ストローク色がトレース色を上書きするため、すべてのマーカーの輪郭が緑色になりました。 

ソリッドのマーカーは黒で、緑の縁取りが施されています。 

中空マーカーの内部を不透明にすることができます。 

Modify Trace Appearance ダイアログで、Fill チェックボックスをオンにします。 

このコマンドは： 

ModifyGraph opaque=1 

 

デフォルトでは、不透明な中空マーカーの内部の色は白です。 

内部の色は、Modify Trace Appearance ダイアログで塗りつぶし色を選択するか、ModifyGraph の mrkFillRGB 

キーワードを使って変更できます。 

コマンドを使って塗りつぶしを設定します： 

ModifyGraph mrkFillRGB=(1,52428,26586) 

 

実線マーカーのストローク色または中空マーカーの塗りつぶし色を設定することで、次の例のように同じ結果を得る

ことが可能です。 

トレース色を緑に設定： 

 

ストローク色を黒に設定すると、マーカーは黒の輪郭と緑の内部を持つようになります： 

 

最後に塗りつぶし色を緑に設定し、中空のマーカーが緑で塗りつぶすと、次の結果となります： 

 

これで、塗りつぶしと中空のマーカーは同じ見た目です。 

これは歴史的な特異点です - Igor Pro 9.0以前では塗りつぶし色を白から変更できませんでした。 

Igor 9 以前では、11や 42などの内部にマークがある中空マーカーは、色を塗りつぶすことができませんでした。 

 

テキストマーカー 

組み込みの描画マーカーに加えて、以下のいずれかをテキストマーカーとして使うように Igor に指示することもで

きます： 

⚫ フォントの１つの文字 

⚫ テキストウェーブの内容 

⚫ 数値ウェーブの内容 

フォント内の１つの文字が主に注目されるのは、フォントで利用可能な特殊記号を使いたいが、それが Igor の組み

込みマーカーに含まれていない場合です。 

指定された文字は全てのデータポイントに使われます。 

残りのオプションは、XY プロットに第３の値を表示する方法を提供します。 

例えば、地震の規模と発生日を示すグラフでは、テキストウェーブを使ってテキストマーカーを供給することで、地

震の発生地点も表示できます。 

あるいは、地震の発生地点と日付のグラフでは、数値ウェーブを用いてマグニチュードを表示し、テキストマーカー

を提供することも可能です。 

XY プロットの各データポイントに対して、対応するテキストまたは数値ウェーブのポイントがマーカーのテキスト

を提供します。 

マーカーウェーブは、Xウェーブおよび Yウェーブと同じ数のポイントを持つ必要があります。 

テキストマーカーを作成するには、Markers または Lines and 

Markers 表示モードを選択します。 

次に、Markers ポップアップメニューをクリックし、Text ボタ

ンを選択します。 

これにより Text Markers サブダイアログが表示され、テキスト

のソースやマーカーのフォント、スタイル、回転などのプロパテ

ィを指定できます。 

すべてのテキストマーカーを同じ量だけオフセットおよび回転さ

せることはできますが、個々のデータポイントに対してオフセッ

トと回転を設定することはできません。 

その場合はタグを使ってください。 
 

データポイント上に文字マーカーを正確に中心に配置するには、X

および Yオフセットを何度か調整する必要がある場合がありま

す。 

例えば、テキストを各データポイントの真上に中央揃えにするに

は、Anchor のセクションで Middle Bottom を選択し、Y Offset 

を 5～10ポイントに設定します。 

一部の項目を他の項目とは異なる方法でオフセットする必要があ

る場合は、タグを使う必要があります（ヘルプ Tags を参照）。 

 

Igorは、マーカーサイズ（Modify Trace Appearance ダイアログで設定）からテキストマーカーに使うフォントサ

イズを決定します。 

使われるフォントサイズはマーカーサイズの 3倍です。 

テキストマーカーと通常の描画マーカーの両方を表示したい場合があります。 

そのためには、グラフ上でウェーブを２回表示する必要があります。 

グラフを作成し、描画マーカーを使うようにトレースを設定した後、Graph→Append Traces to Graph を選択し

て、ウェーブの２つ目のコピーを追加します。 

このコピーにはテキストマーカーを使うように設定します。 

 

矢印マーカー 

矢印マーカーは、例えば、流れや勾配場を示すベクトルプロットの作成に使用できます。 

矢印マーカーはかなり特殊な用途で、少し事前準備が必要となります。 

非常に簡単な例を以下に示します： 

// XY データの作成 

Make/O xData = {1, 2, 3}, yData = {1, 2, 3} 

Display yData vs xData    // グラフの作成 

ModifyGraph mode(yData) = 3    // マーカーモード 

// 各ポイントの長さと角度を制御するための矢印データウェーブを作成 

Make/O/N=(3,2) arrowData    // 矢印の長さと角度を制御 

Edit /W=(439,47,820,240) arrowData 

// arrowData にいくつかデータを投入 

arrowData[0][0]= {20,25,30}       // 列 0: ポイント単位の矢印の長さ 

arrowData[0][1]= {0.523599,0.785398,1.0472} // 列 1: ラジアン単位の矢印の角度 

// トレースを矢印モードに設定して矢印を表示 

ModifyGraph arrowMarker(yData) = {arrowData, 1, 10, 1, 1} 

// RGB カラーウェーブを作成 

Make/O/N=(3,3) arrowColor 

Edit /W=(440,272,820,439) arrowColor 

// カラーウェーブにいくつかの色を格納 

arrowColor[0][0]= {65535,0,0}   // 赤 

arrowColor[0][1]= {0,65535,0}   // 緑 

arrowColor[0][2]= {0,0,65535}   // 青 

// f(z) モードとして色を設定 

ModifyGraph zColor(yData)={arrowColor,*,*,directRGB,0} 

 

詳細については、ModifyGraph コマンドにおける arrowMarker キーワードの説明を参照してください。 

 

線のスタイルと太さ 

Lines between points、Lines and markers、または Cityscape

モードを選択した場合、線のスタイルも選択できます。 

Line セクションの Style 項目を使用して、破線のパターンを変更

できます。 

Markers モード以外のモードでは、線の太さを設定できます。 

線の太さはポイント単位で指定でき、小数点以下の値も指定可能

です。 

線の太さがゼロの場合、線は表示されなくなります。 
 

詳細は、ヘルプ Dashed Lines を参照してください。 

塗りつぶし 

Bars と Fill to zero モードにおけるトレースに対して、Igor は

塗りつぶしタイプの選択肢を提供します。 

塗りつぶしタイプは、透明を意味する None、白で不透明な塗り

つぶしを意味する Erase、Solid、または３種類のグレーパターン

から選択できます。 

パレットからパターンを選択することもでき、正方向領域と負方

向領域に対して、それぞれ独立して塗りつぶしの種類と色を選択

できます。 

 

詳細は、ヘルプ Fill Patterns と Gradient Fills を参照してください。 

 

ウェーブフォームを持つ棒グラフ 

Bars モードが通常の連続 X軸（カテゴリ軸ではない）上にプロットされたウェーブフォームに使われる場合、X値

はウェーブの Xスケーリングから計算されます。 

バーは、特定のポイントの X値から、次のポイントの X値（その値自体は含まない）まで描画されます。 

このようなバーは、通常ヒストグラムにおいて２つの X値の間に収まるカウント数を示すために用いられるため、

「ヒストグラムバー」と呼ばれます。 

 

バーを X値で中央揃えにしたい場合は、カテゴリプロットを作成してください。 

これは従来の棒グラフに適しています（ヘルプ Category Plots を参照）。 

ただし、ウェーブの X値を調整することで、平らな部分が従来の棒グラフと同様に元の X値を中心として表示され

るようにできます。 

データを実際に変更せずにこれを行う方法の一つは、グラフ上のトレースをバー幅の半分だけオフセットすることで

す。 

ドラッグするだけで済みます。 

あるいは Modify Trace Appearance ダイアログを使って、より正確なオフセットコマンドを生成することもできま

す。 

この例では、バーの幅は 0.5 X単位です： 

ModifyGraph offset(wave0)={-0.25,0} 

XYペアを持つ棒グラフ 

X軸が XY ペアによってコントロールされている場合、各バーの幅は２つの X値によって決定されます。 

１つ目の X値はバーの左端の位置を示し、次の X値はバーの右端の位置を示します。 

１本の棒グラフを想像すると、これは２ポイントの XY ペアを必

要とします。 

１つ目の X値でバーの左端を設定し、次の X値で右端を設定しま

す。 

一般化すると、N本のバーを持つ棒グラフの場合、Xウェーブと Y

ウェーブには N+1ポイントが必要です。 

以下は３つの XYポイントを用いて２本のバーを生成する例です： 

Make/O barX={0,1,3}, barY = {1,2,2} 

Display barY vs barX 

ModifyGraph mode=5, hbFill=4 

SetAxis left 0,* 
 

 

グループ化、積み重ね、モードの追加 

通常 y=0 に描画する４つのモード（Sticks to zero、Bars、Fill to zero、Sticks and markers）では、次のトレー

スの Y値まで描画するバリエーションを選択できます。 

4つのバリエーションモードは以下の通りです：Sticks to next、Bars to next、Fill to next、Sticks&markers to 

next。 

この文脈における「next」とは、Modify Trace Appearance および Reorder Traces ダイアログ内のトレース一

覧において、選択したトレースの直後（下）にリストされているトレースを指します。 

これらの４つのモードのいずれかを選択すると、Igor は自動的に 

Grouping ポップアップメニューから Draw to next を選択しま

す。 

このメニューから Add to next モードと Stack on next モード

も選択できます。 
 

Modify Trace Appearance ダイアログの Grouping ポップアップメニューは、レイヤーグラフや積み上げ棒グラフ

などの特殊効果を作成するために使われます。 

利用可能なモードは None、Keep with next、Draw to next、Add to next、Stack on next です。 

Keep with next はカテゴリプロットでのみ使われ、ヘルプ Category Plots で説明しています。 

Draw to next は、通常 y=0 に描画するモードの動作を変更し、現在のトレースと同じ軸ペアに対してプロットさ

れる次のトレースの Y値に描画するようにします。 

次のトレースの X値は、現在のトレースの X値と同じである必要があります。 

そうでない場合、次のトレースは現在のトレースの下端と一致しなくなります。 

Add to next は、プロットする前に現在のトレースの Y値を次のトレースの Y値に追加するよう Igorに指示しま

す。 

次のトレースも「Add to next」を使っている場合、その追加が最初に実行され、以降も同様です。 

通常 y=0 に描画する４つのモードのいずれかと併用する場合、このモードは Draw to next と同様に動作しま

す。 

Stack on next は Add to next と同様に機能しますが、Y値のバックトラックが許可されません。 

つまり、次のトレースの Y値が正の場合、負の値はゼロとして扱われ、次のトレースが負の場合、正の値はゼロとし

て扱われます。 

以下は、小さな正弦波と大きな正弦波の通常のプロットです： 

 

このバージョンでは、小さな正弦波は Add to next モードに設定されています： 

 

Stack on next を適用すると： 

 

Fill to next モードを使って複数のウェーブをグラフにプロットすることで、レイヤーグラフを作成できます。 

データによっては、次のグループに追加モードを使うこともできます。 

例えば、次の通常のグラフでは、各トレースが地層の厚さを表している可能性があります： 

Fill to next と Add to next モードを使うことで、レイヤーをより理解しやすい形で表示できます： 

 

グループ化モードは次のトレースの同一性に

依存するため、グラフ内のトレースの順序を

調整する必要がある場合があります。 

この操作は Reorder Traces ダイアログで行

えます。 

Graph→Reorder Traces を選択してくださ

い。 

選択したトレースをリスト内で上下にドラッ

グして順序を調整し、適切な位置にドロップ

します。 

  

注記 

様々なグループ化、追加、重ね合わせモードで使うすべてのウェーブは、同じ数のポイントと X方向のスケールを持

ち、同じ軸を使って表示される必要があります。 

 

トレースの色 

トレースの色は、カラーポップアップパレットから選択できま

す。 

色に加え、カラーポップアップの alpha プロパティを使って不透

明度を指定できます。 

アルファ値 1.0ではトレースが完全に不透明になり、アルファ値

0.0では完全に透明になります。  

補助（Z）ウェーブからのトレースのプロパティ設定 

補助ウェーブの値に基づいて、トレースの色、マーカー番号、マーカーサイズ、およびパターン番号をポイントごと

に設定できます。 

補助ウェーブは「Zウェーブ」と呼ばれます。 

これは、他のウェーブがトレース上の各ポイントの X座標と Y座標をコントロールする一方で、Zウェーブが第三の

特性をコントロールするためです。 

例えば、地震の発生位置にマーカーを配置し、それぞれの規模を

示すためにマーカーの大きさを変えることができます。 

地震の深さはマーカーの色で示し、地震の種類は異なるマーカー

の形状で示すことができます。 

トレースのプロパティを Zウェーブの関数として設定するには、

Modify Trace Appearance ダイアログ内の Set as f(z) ボタンを

クリックし、Set as f(z) サブダイアログを表示します。 
 

 

f(z) で色を指定 

色を f(z) として定義する場合、4つのモードがあります：Color 

Table、Color Table Wave、Color Index Wave、Three or Four 

Column Color Wave です。 

Color Mode メニューからモードを選択します。  

Color Table モードでは、トレース上の各データポイントの色は、対応する Zウェーブ値を内蔵のカラーテーブルに

マッピングすることで決定されます。 

Log Colors チェックボックスがオンの場合、マッピングは対数関数となります。 

それ以外の場合は線形です。 

Log Colors オプションは、zWaveが数十年にわたる期間をカバーし、より細かい値の変化を詳細に表示したい場合

など有用です。 

zMin および zMax の設定は、Zウェーブの値の範囲をカラーテーブルにマッピングする範囲を定義します。 

範囲外の値は、範囲の末尾の色を継承します。 

zMinまたは zMaxで Auto を選択した場合、Igor は Zウェーブ内で見つけた最小値または最大値を使います。 

Z値のいずれかが NaN の場合、Igor はそれらのデータポイントを Xまたは Yデータが NaNの場合と同様に扱い

ます。 

これはメインダイアログの Gaps 設定に依存します。 

Color Table Wave モードは、Color Table モードと同様ですが、内蔵のカラーテーブルではなく、ユーザーが提供

するカラーテーブルウェーブによって色が決定される点が異なります。 

詳細はヘルプ Color Table Waves を参照してください。 

Color Index Wave モードでは、トレース上のデータポイントの色は、選択した Zウェーブの値を、選択した３列

カラーインデックスウェーブの X軸スケーリングに線形または対数的にマッピングすることで導出されます。 

これは、ModifyImage cindex が画像値（Zウェーブ値の代わりに）を３列のカラーインデックス行列内の色にマッ

ピングする方法に似ています。 

詳細はヘルプ Direct Color Details を参照してください。 

f(z) で色を指定する例 

以下のコマンドは Color Table モードを使い、Zウェーブに基づい

てトレースのポイントを着色します。 

Make/N=5 yWave = {1,2,3,2,1} 

Display yWave 

ModifyGraph mode=3, marker=19, msize=5 

Make/N=5 zWave = {1,2,3,4,5} 

ModifyGraph zColor(yWave)={zWave,*,*,YellowHot} 

 

代わりに３列のウェーブを作成し、RGB値を入力するように編集す

ると： 

Make/N=(5,3) directColorWave 

Row directColorWave[][0] directColorWave[][1] directColorWave[][2]  

 0 1 1  

0 0 0 0 Black 

1 65535 0 0 Red 

2 0 65535 0 Green 

3 0 0 65535 Blue 

4 65535 0 26214 Hot Pink 

このウェーブを使ってマーカーの色を直接コントロールできます： 

ModifyGraph zColor(yWave)={directColorWave,*,*,directRGB} 

 

 

f(z) でマーカーサイズを指定 

Marker size as f(z) は、カラーテーブルモードにおける Color as f(z) と同様に機能しますが、Z 値が min と 

max のマーカー設定で定義したマーカーサイズの範囲にマッピングされる点が異なります。 

この例では、マーカーサイズを関数として第３の値を示していま

す： 

Make/N=100 xData,yData,zData 

xData=enoise(2); yData=enoise(2); zData=exp(-

 (xData^2+yData^2)) 

Display yData vs xData; ModifyGraph mode=3,marker=8 

ModifyGraph zmrkSize(yData)={zData,*,*,1,10} 

 

 

 

f(z) でマーカー番号を指定 

Marker Number as f(z) モードでは、各データポイントの実際のマーカー番号を含む Z ウェーブを作成する必要が

あります。 

マーカー番号コードについては、「マーカー」のセクションを参照してください。 

軸の単位を使った f(z) でマーカーの大きさを指定 

Igor Pro 9.0 以降では、マーカーサイズを軸の単位で指定できます。 

例： 

Make/O xData2 = {1,2,3} 

Make/O yData2 = {100,200,300} 

Make/O zData2 = {10, 20, 50} 

Display/W=(400,50,700,300) yData2 vs xData2 

// マーカーモード、塗りつぶし円マーカー 

ModifyGraph mode=3, marker=19, msize=6 

ModifyGraph grid=1 

ModifyGraph nticks(left)=10 

SetAxis left,0,400 

SetAxis bottom,0,4 

ModifyGraph zmrkSize(yData2)={zData2,*,*,6,10,left} 
 

最後のコマンドは、マーカーサイズをウェーブデータ zData2 によってコントロールするように設定します。 

このデータの値は左軸に合わせてスケーリングされます。 

例えば、zData2[2] は 50であり、yData2のポイント 2のマーカーの半径を左軸上で 50単位に設定します。 

指定された軸はグラフ上の任意の軸で構いません。 

その軸が後で削除された場合、マーカーサイズは「軸なし」状態に戻り、軸を指定していない状態となります。 

軸の単位でマーカーサイズをスケーリングする場合、zmrkSize キーワードの zMin、zMax、mrkMin、mrkMax パ

ラメータに指定した値は、マーカーサイズに影響を与えません。 

ただし、指定した軸を削除した場合、これらのパラメータがトレースのマーカーサイズをコントロールします。 

マーカーサイズのスケーリングが、トレースがプロットされている軸のスケールとは独立している場合、そのスケー

ルを表示するために自由軸を使用できます。 

これを行うには、目的の範囲を持つ自由軸を作成し、それを zmrkSize キーワードの軸として設定します。 

自由軸に関する説明は「軸の種類」のセクションを参照してください。 

指定した軸が対数軸の場合、Z値の解釈が変化します。 

対数軸と同じポイント長を持つ概念的な軸を想像してください。 

概念軸の最小値と最大値を、対数軸の最小値の対数と最大値の対数に概念的に設定します。 

特定の Z値に対するマーカーサイズは、概念軸上の Z単位の長さです。 

これを視覚化するために、この概念軸を表す自由軸を作成できます。 

 

f(z) でパターン番号を指定 

Pattern Number as f(z) モードでは、各データポイントの実際のパターン番号を含む Z ウェーブを作成する必要

があります。 

パターン番号の一覧については、「塗りつぶしパターン」のセクションを参照してください。 

 

f(z) による色の指定の凡例の例 

色を f(z) モードとして使うグラフがある場合、各色が何を表すかを識別する凡例を作成することをお勧めします。 

このセクションでは、この目的のために凡例コマンドの機能を使う方法を示します。 

以下のコマンドを、一つずつ実行してください： 

// テストデータの作成 

Make /O testData = {1, 2, 3} 

// マーカーモードでグラフを表示 

Display testData 

ModifyGraph mode=3,marker=8,msize=5 

// シンボルがトレースから取得される通常の凡例を作成 

Legend/C/N=legend0/J/A=LT "\\s(testData) First\r\\s(testData) Second\r\\s(testData) 

Third" 

// マーカーの色をコントロールするためのカラーインデックスウェーブを作成 

Make /O testColorIndex = {0, 127, 225} 

// グラフのトレースを f(z)モードで色を使うように変更 

// Rainbow256 は組み込みカラーテーブルの名前 

// 数値 0と 255は、カラーテーブルの最初と最後のエントリに対応するカラーインデックス値を設定 

ModifyGraph zColor(testData)={testColorIndex,0,255,Rainbow256,0} 

// 凡例が色を表示するように変更 

Legend/C/N=legend0/J/A=LT "\\k(65535,0,0)\\W608 Red\r\\k(0,65535,0)\\W608 

Green\r\\k(0,0,65535)\\W608 Blue" 

結果は次のグラフです： 

 

最後のコマンドでは、凡例で使うマーカーを指定するために \Wエスケープシーケンスを使い（この場合は円形マー

カーの 08）、また、マーカーの太さも指定しています（6は 1.0ポイントを意味します）。 

\k エスケープシーケンスは、\W で指定されたマーカーのストロークに使う色を指定します。 

マーカーの塗りつぶし色を指定するには \K を使います。 

色は RGB形式で指定され、各色成分は 0から 65535の範囲に収まります。 

この例では、１つのバックスラッシュは Igor リテラル文字列においてエスケープ文字として扱われるため二重バッ

クスラッシュを使っています。 

最終テキストにバックスラッシュを含める必要があるため（Igor が \k および \W にこれを要求するため）、リテ

ラル文字列では二重バックスラッシュを使う必要があります。 

Add Annotation ダイアログに凡例テキストを入力する場合、１つのバックスラッシュのみを使います。 

するとダイアログは必要なコマンドを生成し、そのコマンドには二重バックスラッシュが含まれます。 

トレースのオフセット 

グラフ内のトレースを、関連するウェーブデータの変更なしに、水平方向または垂直方向にオフセットできます。 

これは主に、トレースの形状を比較したり、重なり合ったトレースを広げて見やすくしたりするために有用です。 

 

各トレースには Xオフセットと Yオフセットがあり、初期値はいずれもゼロです。 

Modify Trace Appearance ダイアログで Offset チェックボックスをオンにすると、トレースの Xオフセットと Y

オフセットを入力できます。 

グラフ上でクリックしてドラッグすることでオフセットを設定す

ることもできます。 

これを行うには、オフセットしたいトレースをクリックします。 

マウスを約 1秒間押し続けてください。 

グラフの左下隅に読み取りボックスが表示されます。 

読み取り値は、トレースをドラッグする時に X軸と Y軸のオフセ

ットを表示します。 

指定されたトレースの表示に時間がかかりすぎなければ、画面上

でトレースをドラッグしながらその形状を確認できます。  

ウェーブをオフセットする時に Shift キーを押すと、Igor はオフセットを水平方向または垂直方向に制限します。 

Caps Lock キーを押すとトレースドラッグを無効にできます。 

トラックボールユーザーには便利な機能です。 

誤って不要な場所にトレースをドラッグした場合は、Edit メニューから Undo を選択してください。 

トレースにタグを添付し、その現在のオフセット値を表示することが可能です。 

詳細はヘルプ Dynamic Escape Codes for Tags を参照してください。 

グラフで自動スケーリングが有効な場合、Igor はトレースオフセットを考慮しようとします。 

軸のスケーリングに影響を与えずにトレースのオフセットを設定したい場合は、Graphs メニューの Set Axis 

Range 項目を使って自動スケーリングを無効にしてください。 

対数軸を使うトレースをオフセットする場合、Igor は軸が対数でない場合と同じ距離だけトレースを移動します。 

トレースの形状は変化せず、単に移動されます。 

ウェーブデータに定数を加えてトレースをオフセットしようとすると、トレースが歪んでしまいます。 

トレースの乗算器 

トレースを相殺するだけでなく、トレースをスケーリングするための乗数も指定できます。 

プロットに使われる実効値は、「乗数×データ+オフセット」となります。 

デフォルト値のゼロは、乗数が指定されていないことを意味し、データをゼロで乗算すべきことを意味するものでは

ありません。 

通常の（対数ではない）軸では、トレースのオフセットで説明したのと同じクリック＆ホールド操作で、トレースを

インタラクティブにスケーリングできます。 

まず、基準点として機能させるため、カーソル Aをトレース上の任意の位置に配置する必要があります。 

次に、トレースをクリックして押し続けることでオフセットモードに入った後、Altキーを押すとオフセットではな

く乗数を調整できます。 

キーを押したり離したりすることで、スケーリングとオフセットを切り替えることができます。 

対数軸では、トレース乗算器は対数軸上のトレースをオフセットする代替手段を提供します（覚えておいてくださ

い：log(a*b) = log(a) + log(b)）。 

互換性の理由と、トレースオフセット方式が対数軸モードと直線軸モードの切り替えをより適切に処理するため、

Igor はトレースをドラッグする時に、オフセットがゼロで乗数がゼロでない場合（デフォルトは「設定なし」を意

味します）にのみ乗数方式を使います。 

したがって、乗数のテクニックを使うには、コマンドラインまたは Modify Trace Appearance ダイアログのオフセ

ットコントロールを使って、ゼロ以外の乗数を設定する必要があります。 

この目的には 1が適切な設定です。 

 

トレースを隠す 

グラフ内のトレースを非表示にするには、Modify Trace 

Appearance ダイアログで Hide Trace チェックボックスをオン

にします。 

トレースを非表示にした場合、Include in Autoscale チェックボ

ックスを使って、グラフのオートスケール時に非表示にしたトレ

ースのデータを使うかどうかをコントロールできます。  

 

 

複素数表示モード 

複素数データのトレースを表示する時には、Modify Trace 

Appearance ダイアログの Complex Mode ポップアップメニュ

ーを使って、データの表示方法をコントロールできます。 

複素数成分と実数成分を一緒に表示することも個別に表示するこ

ともでき、また、実数成分（絶対値）や位相を表示することもで

きます。 

デフォルトの表示モードは Lines between points です。 

波形の実部と虚部を点ごとに並べて表示するには、Sticks to zero 

モードを使います。 

 

ギャップ 

Igor では、ウェーブ内の欠損値または未定義値は浮動小数点値 NaN（「Not a Number／数値ではない」）として

保存されます。 

通常、Igor はグラフ上で NaN を空白として表示し、その点にデータが存在しないことを示します。 

この設定は Modify Trace Appearance ダイアログの Gaps チェ

ックボックスでコントロールできます。 

このチェックボックスがオンの場合（デフォルト）、Igor は欠損

値をデータのギャップとして表示します。 

 

チェックボックスのチェックを外すと、Igor は欠損値を無視し、

欠損値の両側にある利用可能なデータポイントを接続します。 

 

エラーバー 

Modify Trace Appearance ダイアログの Error Bars チェックボ

ックスは、選択したトレースにエラーバーを追加します。 

このチェックボックスを選択するか、Options ボタンをクリック

すると、Igor は Error Bars サブダイアログを表示します。 

エラーバーは、グラフ内のトレースに追加できるスタイルです。 

誤差値は、定数、所定のポイントにおけるウェーブの値の固定割

合、所定のポイントにおけるウェーブの値の平方根、または他の

ウェーブから取得した任意の値とすることができます。 

この最後のケースでは、誤差値は上方向、下方向、左方向、右方

向ごとに個別に設定できます。 
 

Y Error Bars と X Error Bars のポップアップメニューから、目的のモードを選択してください。 

選択したモードによってダイアログが変化します。 

% of base モードでは、ベースウェーブのパーセンテージを入力します。 

sqrt of base モードでは、これ以上の値を入力する必要はありません。 

このモードは、データがカウント値である場合にのみ意味を持ちます。 

constant モードでは、X 方向または Y 方向の定数誤差値を入力します。 

+/- wave モードでは、正と負の誤差値を供給するウェーブを選択します。 

+/- wave を選択すると、ポップアップメニューが表示され、そ

こから上下または左右の誤差値を供給するウェーブを選択できま

す。 

これらのウェーブは誤差ウェーブと呼ばれます。 

誤差ウェーブの値は、各ポイントからその誤差バーまでの線の長

さを指定するため、通常すべて正であるべきです。 

これは片側誤差バーをサポートする唯一のモードです。 

誤差ウェーブは、ベースウェーブと同じ数値タイプや長さを必要

としません。 

エラーウェーブにおける、あるポイントの値が NaNの場合、その

ポイントに対応するエラーバーは描画されません。 

Cap Width 設定は、エラーバーの端にあるキャップの幅をポイン

ト単位の整数で設定します。 

キャップ幅を auto（またはゼロ）に設定することも可能です。 

この場合、Igor はグラフのサイズに適したキャップ幅を選択しま

す。  

また、キャップの幅はマーカーのサイズの 2倍に 1を加えた値に設定されます。 

最適な結果を得るには、キャップの幅は奇数にする必要があります。 

キャップの太さとエラーバーの太さを設定できます。 

これらの設定の単位はポイントです。 

これらは小数点以下の数値とすることができます。 

非整数の時の太さは、エクスポートまたは印刷時に適切に生成されます。 

キャップの太さをゼロに設定すると、キャップは描画されません。 

XY Error box チェックボックスを有効にすると、不確実性の領域

を示すためにエラーバーではなくボックスが描画されます。 

エラー値の１つ以上が NaN であるポイントについてはボックス

は描画されません。 

右はエラーバー付きのグラフの簡単な例です。 

一番上のトレースは +/- Wave モードをい、+ ウェーブのみを

表示しました。 

エラーバーの方向を反転させるため、誤差ウェーブの最終値を負

にしました。 

 

エラーシェーディング 

Igor Pro 7 以降では、エラーバーの代わりにシェーディングを使

用できます。 

Error Bars サブダイアログで、Use shading instead of bars チ

ェックボックスをオンにしてシェーディングを有効にします。 

シェーディングモードは、+error レベルから -error レベルの

間、または +error からデータ、データから -error までの間で

塗りつぶします。 
 

これは shade キーワードと共に使われるパラメータによって異なります。 

以下のコマンドは、透明度を用いた多重かつ重なり合うエラーシェーデ

ィングを実例で示しています： 

Make/O/N=200 jack=sin(x/8), sam=cos(x/9) + (100-x)/30 

Display /W=(64,181,499,520) jack,jack,sam 

ModifyGraph lsize(jack)=0,rgb(sam)=(0,0,65535) 

ErrorBars jack shade={0,0,(0,65535,0,10000),(0,0,0)},const=2 

ErrorBars jack#1 shade={0,0,(0,65535,0,10000),(0,0,0)},const=1 

ErrorBars sam shade={0,0,(65535,0,0,10000),(0,0,0)},const=1 

シェーディングは、旧式の GDIグラフィック技術では機能しません。 

詳細はヘルプ Graphics Technology を参照してください。  

次のコマンドは、異なる +error および -error のシェーディングと、

パターンの使用例を示しています： 

Make/O jack=sin(x/8) 

Display /W=(64,181,499,520) jack 

ErrorBars jack 

shade={0,73,(0,65535,0),(0,0,65535),11,(65535,0,0),(0,65535,0)},cons

t=0.5 

詳細については、ErrorBars コマンドの shade キーワードを参照して

ください。 

 

エラー楕円 

Igor Pro 9.0 以降では、エラーバーやエラーボックスの代わりにエラー楕円を指定できます。 

エラー楕円は X方向と Y方向の誤差を両方表示し、Xと Yの相関関係も示します。 

エラー楕円のデータは、各データポイントごとに 1行を持つ 3列のウェーブデータで提供する必要があります。 

このウェーブは、エラーバーダイアログで Ellipse Wave とラベル付けされています。 

ErrorBars コマンド内では ewave=ew で識別されます。 

ew の各行は、１つのデータポイントに対するエラー楕円の情報を含みます。 

ew の列の解釈は、ErrorBars ダイアログで Data Type と表示され、ErrorBars コマンドの ELLIPSE キーワード

の mode パラメータで識別されるモードに依存します。 

mode = 0: ew には、Xの標準偏差、Yの標準偏差、および Xと Yの相関が含まれます。 

mode = 1: ew には、Xの分散、Yの分散、および Xと Yの共分散が含まれます。 

ErrorBars コマンド（ew）で指定する場合、各トレースのデータポイントに対して 3列と 1行からなる実質的な 2

次元ウェーブとなる限り、部分範囲指定を含めることができます。 

「サブ範囲表示構文」のセクションを参照してください。 

 

エラー楕円の色 

各エラー楕円の色は、その楕円が結び付けられているデータポイントの色から取得されます。 

通常、それは単なるトレースの色です。 

個々のデータポイントとその対応する楕円の色は、トレースカラーを f(z) として設定するか、ポイントのカスタマ

イズ（「ポイントのカスタマイズ」のセクションを参照）によって変更できます。 

Error Bars ダイアログの Fill Alpha 設定を使うと、指定したトレ

ースのすべてのエラー楕円の不透明度を 0（完全に透明）から 1.0

（完全に不透明）の範囲で指定できます。 

これは、すべてのエラー楕円の塗りつぶし色の透明度を指定し、

データポイントの透明度を上書きします。 

ErrorBars コマンドでは、ELLIPSE キーワードを使って、アルフ

ァ値を 0（完全透明）から 65535（完全不透明）までの値で指定

します。 

アルファに関する背景情報については、ヘルプ RGBA Values を

参照してください。 

 

エラー楕円の例 

次の例はエラー楕円の使い方を示しています。 

プロシージャウィンドウに次を入力して、コンパイルします。 

Function ErrorEllipseDemo() 

 Make/O/N=5 data = 5*sin(x) 

 Display data 

 ModifyGraph mode=3, marker=19 // マーカーモード、塗りつぶした円マーカー 

 Make/O/N=(5,3) ellipseData  // エラー楕円パラメーター 

 // 列 0: X における標準偏差 

 ellipseData[0][0]={0.944527,1.19611,0.32858,1.64494,1.71742} 

 // 列 1: Y における標準偏差 

 ellipseData[0][1]={0.357613,0.680198,0.665561,0.581878,0.948245} 

 // 列 2: Xと Yの相関係数。範囲は[-1, 1] 

 ellipseData[0][2]={0.562396,0.707809,0.396551,-0.516456,0.554322} 

 ErrorBars data, ELLIPSE={0, .95, .3*65535}, ewave=ellipseData 

 // ポイント２を青にする 

 ModifyGraph rgb(data[2])=(0,0,65535) 

End 

コマンドラインで次を実行します。 

ErrorEllipseDemo() 

ポイントでのカスタマイズ 

グラフ上で表示されるトレース上の個々のポイントの外観を、

棒、マーカー、点、線からゼロモードまでカスタマイズできま

す。 

これを対話的に行うには、トレース上の目的のポイントを右クリ

ックし、コンテキストメニューから Customize at Point→

Customize at Point を選択します。 

トレースリストにエントリが表示され、そのポイント番号が角括

弧で囲まれて表示された状態で、Modify Trace ダイアログが表示

されます。 
 

このような項目が選択されると、ダイアログにはカスタマイズ可能なプロパティのみが表示されます。 

ポイント上で右クリックし、Customize at Point→Remove Customization Here を選択すると、そのポイントの

カスタマイズを削除できます。 

トレースからすべてのカスタマイズを削除するには、トレースを右クリックし、Customize at Point→Remove 

Customizations from Trace を選択します。 

